
ESCI 386 – IDL Programming for Advanced Earth Science Applications 

Lesson 5 – Main Programs, Procedures, and Functions 

 

Reading:  Bowman, Chapters 15 

 

MAIN PROGRAMS 

• An IDL main program consists of a series of IDL statements, concluding with the END 

statement. 

• Main programs differ from scripts in a couple of ways: 

o A script doesn’t have an END statement. 

o A script can’t contain blocks of code such as loops and if-then statements. 

• Main programs are compiled and ran from the IDL command line by typing 

.r program_name.pro 

• Once a main program is compiled, it can be reran by just typing .g at the IDL 

command line. 

• When a main program is ran from the command line, all the variables defined within 

the main program itself (but not those in any called procedures or functions) remain 

available after the program is ran. 

• As an example of a main program, type the following into a text file and save it as 

“hello.pro” (you don’t need to type the comments, which follow the semicolons. 

fn = “”  ;  creates a string variable for the first name 

ln = “”  ;  creates a string variable for the last name 

read, fn, prompt = “Enter your first name:” 

read, ln, prompt = “Enter your last name:” 

print, “Hello, ” + fn + “ ” + ln + “.” 

END 

• This program prompts the user to enter their first and last names, which are placed 

into the string variables fn and ln respectively.  I then prints a greeting. 

• Run the program from the IDL command line by typing .r hello_name.pro 

• If you want to rerun the program, you could type .r hello_name.pro again.  

However, since the program has already been compiled, you could just type  .g . 

• After running the program, type print, fn  from the command line, and notice that 

IDL remembers what this variable is even after the program is finished executing.  This 

is handy, because you can still run IDL interactively and have access to all the variables 

that were used in the main program. 

 

PROCEDURES 

• A procedure is an IDL program that can be called either from the command line or 

from within another IDL program. 

• A procedure is analogous to a subroutine in Fortran. 

• Features of procedures are: 

o Must begin with a PRO statement followed by the name of the procedure. 

o Must end with an END statement. 

o You can pass arguments to and from procedures (but, you don’t have to). 

• Procedures are called by their name followed by a comma and any arguments (using 

commas to separate the arguments). 

• Both incoming arguments, and any return arguments, are passed in the argument list 

and must be included in the procedure call statement. 

• A procedure does not have to have return arguments. 

• When IDL encounters a call to a procedure it will automatically search for a file having 

the same name as the procedure, and then compile that file. 

o This is very unlike FORTRAN, where you would have to compile and link all 

subroutines before running. 

• We can write our greeting program that we originally wrote as a main program above 

as a procedure instead. 

• Type the following into a file and save it as “hello_proc.pro” 

PRO HELLO_PROC, first, last, return_string 

return_string = "Hello, " + first + " " + last + "." 

END 

• Then, at the command line type 

hello_proc, “Joe”, “Sixpack”, greeting 

print, greeting 

• We passed the strings “Joe” and “Sixpack” into the procedure, and the procedure 

returned the string “Hello, Joe Sixpack.” in a variable called greeting. 

• Now from the command line type print, first 

• You get a message saying that this variable is undefined!  This is because the variables 

used within a procedure are local to that procedure, and aren’t available outside of it. 

o Some IDL programmers will say that you shouldn’t write main programs, but 

should instead do everything as a procedure or function.  I disagree.  I like writing 

main programs when possible because the variables are still available to you at the 

command line when the main program is finished executing. 

• You can turn any main program into a procedure just by including the PRO statement. 

o Type .reset into the command line to clear the variable space.  Then: 

o Add the statement PRO HELLO to the beginning of your hello main program from 

the previous section, and then call it by typing hello at the command line. 

o Notice that it acts just like the main program did, however, the variables fn and ln 

aren’t available to you at the command line (unless you didn’t reset, in which case 

they may still be there from when you ran the main program earlier). 

 

FUNCTIONS 

• A function is similar in some respects to a procedure, but ALWAYS returns a value. 

o If the user doesn’t write the function to return a value, the function returns a 

default value of zero. 

• An IDL function is analogous to a Fortran function. 

• Features of functions are: 

o Must begin with a FUNCTION statement followed by the name of the procedure. 

o Uses a RETURN statement telling the function what value to return to the calling 

program. 

� Unlike procedures, the return value of a function is not included in the argument 

list. 

� If the RETURN statement is omitted, the function returns a default value of zero. 

o Must end with an END statement. 

• Functions are called by their name followed by any arguments in parenthesis (using 

commas to separate the arguments). 

o If a function does not expect any arguments, then it is called with empty 

parentheses. 

• When IDL encounters a call to a function it will automatically search for a file having 

the same name as the function, and then compile that file. 

• We can write our greeting main program as a function as follows (save it in a file 

named “hello_func.pro”: 

FUNCTION HELLO_FUNC, fn, ln 

return, "Hello, " + fn + " " + ln + "." 

END 

• We can call the function by typing print, hello_func(“Joe”, “Sixpack”) at 

the command line. 

• Again, notice that the variables fn and ln aren’t available to you at the command line 

(unless you didn’t reset). 

• To demonstrate the calling of procedures and functions from within a main program, 

write a main program that looks like the following, and execute it from the command 

line: 

fn = ""      ;  creates a string variable for the first name 

ln = ""      ;  creates a string variable for the last name 

read, fn, prompt = "Enter your first name: " 

read, ln, prompt = "Enter your last name: " 

hello_proc, fn, ln, g 

print, "Results from procedure: ", g 

h = hello_func(fn, ln) 

print, "Results from function: ", h 

END 

 

NAMING FUNCTIONS AND PROCEDURES 

• Functions and procedures should be stored in file that has the same name as the 

function or procedure, only with a “.pro” as the file extension. 

o A procedure called PRO HEY_THERE would be saved in a file called 

“hey_there.pro”. 

o A function called FUNCTION NICE_CAR would be saved in a file called 

“nice_car.pro”. 

• The function or procedure name itself is NOT case sensitive 

o i.e., PRO HEY_THERE, PRO hey_there, and pro HeY_tHeRe are all equivalent 

in the file itself. 

• The file name IS case sensitive, and should ALWAYS be lower case. 

o “hey_there.pro”, and not “HEY_THERE.PRO” 

 

HOW PROCEDCURES AND FUNCTIONS TREAT VARIABLES PASSED TO THEM 

• When a scalar variable is passed to a function or procedure, IDL just passes a reference 

to the address in memory where that variable’s value is stored.  IT IS IMORTANT TO 

REALIZE THAT IDL DOES NOT PASS A COPY OF THE VARIABLE! 

o If the procedure or function changes the value of a variable, its value in the 

calling program or procedure also changes. 

EXAMPLE:  The function below takes in two arguments, x, and y, and multiplies 

them, returning the value of their product and putting it back into x.  Type the 

function into a text editor and save it as multiply.pro 

FUNCTION multiply, x, y 

  x = x*y 

  return, x 

END 

Then from the IDL command line type 

a = 2 

b = 6 

c = multiply(a,b) 

print, a, b, c 

The result of the print is 12  6  12, not 2  6  12 as you might expect. 

Notice that the value of a (which was passed into the variable x in the function) was 

altered by the function, and was therefore altered in the main program that called 

the function. 

• This behavior results from the fact that IDL passes scalar variables by referencing (or 

pointing) to their memory addresses. 

o The variable a in the main program and the variable x in the function both 

point to the same location in memory!  So  do the variables b and y!  The 

diagram below illustrates this. 

 

• IDL does this because it saves time and memory by just passing the address to the 

memory to the function, rather than creating a duplicate copy of the variable.  

However, if you aren’t expecting this behavior it can cause problems for you. 

• If you want to preserve the value of a variable that you are passing to a function or 

procedure, then you should copy it to a new variable and then call the procedure with 

the new variable. 

EXAMPLE:   

FUNCTION multiply, x, y 

  x = x*y 

  return, x 

END 

From the IDL command line type 

a = 2 

b = 6 

d = a 

c = multiply(d,b) 

print, a, b, c, d 

The result of the print is 2  6  12  12.  The variable d got changed, but we 

don’t care because it was a throwaway variable that we weren’t worried about 

preserving in the main program. 

• How IDL handles arrays passed to functions and procedures depends on whether the 

array is passed with subscripts or not. 

o If an entire array is passed, without using any subscripts, than IDL passes the array 

by reference just like for a scalar variable, and the array might be altered within the 

procedure or function. 

o If subscripts are used when passing either an entire array or a partial array, then 

IDL passes the value or values of the array elements.  In this case, the original array 

won’t be changed within the procedure or function. 

• For example, if we do 

a = indgen(5) 

b = 2  

c = multiply(a,b) 

print, a 

We get  =>  0, 2, 4, 6, 8 

but if instead we do 

c = multiply(a[*],b) 

print, a 

we get  => 0, 1, 2, 3, 4 

o The difference between the two calls to the function above is that in the first the 

array was passed without subscripts, while in the second it was passed with 

subscripts.  In both cases the entire array was passed, but with very different 

results. 

• To quote the IDL online help: 

o Expressions, constants, system variables, and subscripted variable references are 

passed by value. 

o Variables are passed by reference. 

• WARNING:  Because IDL sometimes passes arguments by reference, and at other times it 

passes them by value, it is best to avoid directly changing the value of an input variable 

within a function or procedure, unless you are certain that it will have no ill effects in the 

calling program. 

o If you need to alter the value of an input variable, you should copy it to another 

variable first, and modify the copy of the variable. 

 

USING KEYWORDS IN FUNCTIONS AND PROCEDURES 

• IDL allows the use of keywords in functions and procedures. 

• Keywords allow more flexibility in the function or procedure. 

• An example of the use of a keyword would be a function that converts temperature in 

degrees Celsius into either Fahrenheit or Kelvin.  The example below shows such a 

function: 

FUNCTION CELSIUS_CONVERT, deg_c, kelvin=kelvin 

  if keyword_set(kelvin) then begin 

    return, deg_c + 273.15 

 end 

 return, deg_c*9.0/5.0 + 32.0 

END 

If this function is called without the keyword /kelvin being set, then it will return the 

value in Fahrenheit.  If the keyword /kelvin is set, then the returned value will be in 

Kelvin.  Try the following examples and see what you get. 

print, celsius_convert(5) 

print, celsius_convert(5, /kelvin) 

• The KEYWORD_SET() function returns a 1 if the keyword is set, and a 0 if the keyword 

is not set. 

• You are not limited to one keyword per function or procedure.  You can use as many as 

are needed. 

 


